High Performance, Low Cost New Printer

By Owen F. Thomas

An electric discharge printer is being offered for sale to the OEM market for $257 assembled. This could be the output device that the hobby and small systems users have been wishing for. The printer will produce 44 characters in half a second in a 5x7 dot matrix format on paper 6 cm. wide; and it will do that continuously. Software control will produce larger characters for emphasis and will allow graphics to be printed. The low cost is achieved by a simple printing mechanism, limited number of components on the interface board, and software to provide all of the logic control to the printer.

There is no distributor to the hobby market at the present, but the manufacturer will deliver single units from the factory. You must write to Electronic Products Associates, Inc., 1157 Vega Street, San Diego, CA 92110 if you want one at once. Perhaps someone will advertise soon as a distributor to the hobby and small system users.

The printer mechanism comes from a distributor in Germany. Those who wish to try assembling their own system can contact Datamega KA, 8011 Putzbrunn, Munich, Germany. Their catalog lists the printer mechanism only as MP 310/21L for 160 German Marks. I do not know of a source for single units in this country, but anyone wishing to build their own interface from information in this article can try offering the equivalent of 160 German Marks to EPA. A cash offer would probably not be refused. Printed circuit board layout, or wire wrapping, must be done carefully in a homebrew system to keep noise out of the low level signal from a magnetic pickup; the other circuitry is straightforward.

All good things have their problems, right? This printer requires a special aluminum coated paper which is available from EPA at 95¢ per 90 foot roll. One roll will hold about 6,000 lines, or about 250,000 characters. This is not ex-



Figure 1. Megatron MP310 Printer Mechanism.



PHOTO 1



PHOTO 2

pensive for systems with video displays which need occasional permanent copies from the screen. The copy is archival, it never fades, and copy machines will work well from the original. So there is some bad news and good news about paper!

HARDWARE DESCRIPTION

The print mechanism from Megatron is a box full of pulleys, cables, gears, motor, reed switch, and magnetic pickup coil for timing pulses. There are fourteen wires from the box to two small paddle boards. One paddle board connects to the motor, reed switch, timing pickup, and paper common electrode. The other paddle board connects via a flexible printed circuit to the seven electrodes on the print head. This makes the print head easy to replace.

A conceptual layout of the functioning parts is shown in Figure 1. The seven printing electrodes are mounted vertically in a small plastic block which is moved in a racetrack path from left to right in contact with the paper and then moved away from the paper for return to start. The reed switch is open for a short space just before the electrodes touch the left edge of the paper. The magnetic pickup generates a periodic voltage from a gear in the drive mechanism while the motor is turning (driven by 24V DC). A carbon roller in contact with the paper provides a common path for the print current when the print electrodes are grounded but limits the current. An automatic line feed ratchet, which is not illustrated, advances the paper one line just after the print head leaves the paper.

COMPUTER INTERFACE CIRCUIT

A minimum of extra components has been provided to allow computer control of motor and print current and to provide timing marks to the computer. The motor is controlled with one output bit, and the print electrodes are controlled with the other seven bits of an 8-bit output word. The timing marks are provided to the computer as low impedance to ground; they can be used to ground pullup resistors on two computer input lines.

Table 1. Control Signal and Power Supply Description

Signal Description	Pin Number	Signal Source	Active Signal
Motor Run	23	Computer bit 7	+ 5v., 1 ma.
Print Electrodes:			
Print top dot	9	Computer bit 0	1 ma. source
2nd dot	11	Computer bit 1	1 ma. source
3rd dot	13	Computer bit 2	1 ma. source
4th dot	15	Computer bit 3	1 ma. source
5th dot	17	Computer bit 4	1 ma. source
6th dot	19	Computer bit 5	1 ma. source
Bottom dot	21	Computer bit 6	1 ma. source
Line end	27	Printer	current sink
Character timing	25	Printer	current sink
Power	43,44, 45,46		40v. ¼ amp.
Power ground	1,2,3,4		

A circuit diagram of the interface is illustrated in Figure 2, and a photograph of the printed circuit board holding the printer and interface components is in Photo 2. The positive input voltage is applied directly to the print head common, and each electrode is provided with its own current path to ground through one of the 2N2222 transistors Q7 through Q13 when the base current is turned on by a high bit output. A printer supply of 40 volts will make a very dark mark on the paper; 30 volts will make an acceptable mark. The paper common is also connected to the printer case, so it must not be grounded or connected to metallic parts of the system.



Figure 2.

The other computer output bit drives the five transistors Q2 through Q6. This complicated circuit is required to provide dynamic braking to the motor when the output bit is high as well as to pass current to the motor when the bit is low. A voltage regulator is provided to produce the specified 24-volt motor drive regardless of what printing voltage is used.

The reed switch is grounded on one side, and the other side goes direct to the pullup resistor on the computer input line. When the reed switch opens, the bit goes high, and the computer can stop the motor before the next line begins.

The periodic timing marks are only about 0.6 volts peak to peak and require special conditioning for computer input. The pickup coil shield is biased up from ground by one diode drop across CR2, making the base of Q1 the same potential as the emitter when the motor is off. When the motor runs, it will cause the pickup coil to turn Q1 on for a part of each timing period and therefore provide a low impedance to ground for a pullup resistor on the computer input line. This will cause the collector of Q1 to be at a low voltage during part of each timing period to generate a timing signal for use by the computer in determining when to activate the print electrode bits.

No more interface components are required to allow the computer software to print a 44-column line as the printer head crosses the paper in half a second. The subroutine which controls this interface will be explained in detail in the context of a software package to print a line of characters which are input from an ASCII keyboard. The printer control software is a single subroutine called LINPTR described under software.

SOFTWARE DESCRIPTION



Figure 3. Table of Software Organization.

This description uses a set of visual aids to enhance the introduction of data and control concepts before presenting program listings.* Figure 3 shows the organization of software modules to accomplish a line typer task. Each module of the figure has a key number in the lower right corner of its box as a key to the following descriptions.

Overview



Figure 4. "Line Typer"

The overview of all software is illustrated in Figure 4. This figure is typical of all the module descriptions; it is divided into three boxes, one to show data input to the module, one to show the process performed by the module, and one to

*The technique is explained in "Systems Design and Documentation" by Harry Katzan, Jr.

show the data output by the module. Open arrows show data movement from input box through the process box to the output box. Solid arrows show where the process starts and what process comes next.

The overview diagram shows general features rather than complete detail. This process accepts input from a keyboard, has five steps in the process, and produces a printed line when the ASCII character from the keyboard is a carriage return. Other ASCII characters are converted to printer control words and placed in a control word buffer. When the process is finished, control will be returned to step 2 or 3 of the same process. This module is therefore a closed loop; it will continue to accept keyboard input and pass it to the printer until it is interrupted by a reset or a timeout from a late timing mark. Each step of this process will be expanded in the following figures and descriptions. After all the modules are understood, it will be easy to use them as they are or to adapt the coding for other applications.

Initialization



Figure 5. "Initialize Printer"

Printer initialization details are presented in Figure 5. Initialization is performed by a single subroutine called INITPR which is written to be contained in a read-only memory at any location which may be convenient to a system designer who wishes to use the printer. The read-only memory can also contain a table of printer control words; the address of the table is determined by the program and stored in random access read/write memory where it will be referred to by the name TBLADR. The peripheral interface adapter, or an 8-bit output port, is set for all bits to be outputs. The motor is started and allowed to run to the start of a new line. At line end it should set a data bit or generate an interrupt; in the event that this does not occur in a reasonable time, there will be a software interrupt generated. This interrupt can be interpreted by the system operator or system software to start necessary corrective action.

If the printer reaches line end in time, the printer is ready to operate, so default values will be stored in three locations to control the format of the printed line. A line generated by these values will have 44 characters separated by 2/5ths character spacing. The number 1 to control spot size, and therefore character size, is stored and will be referred to by the name SPTSIZ. The number 44 to control number of characters is stored and will be referred to by the name CHRCNT. The number 2 to control space between characters is stored and will be referred to by the name BLANKN. These default values could be modified by other software if desired as long as the product CHRCNT (5+BLANKN) is less than 310.

Line typer software control starts with printer initialization; control passes to buffer initialization when the printer responds or goes to system software when the printer does not respond.

Buffer Initialization

Printer control characters are stored in a buffer in read/write memory until the printer is commanded to print a line. This buffer is initialized by the software illustrated in Figure 6. The software is a single subroutine called CAINIT, and it can be in read-only memory when the system designer

has established the buffer location and size. The beginning address of the buffer is stored and will be known by the name BUFSTR. The beginning address is also stored in the location called BUFPTR where it will be used as a pointer for the next empty location in the buffer. The Buffer end address is computed and stored in the location called BUFEND. A control word is stored in all buffer locations to cause the motor to run without printing unless the word is changed when a character is input from the keyboard. This has the effect of causing blank lines to be printed when a carriage return is the first character from the keyboard.



Figure 6. "Initialize Buffer"

Program control comes to this module from printer initialization or from printing a line. Control goes to input an ASCII character from the keyboard.

Input ASCII



Figure 7. "Input ASCII"

The keyboard input is illustrated in Figure 7. All the software for ASCII input from a serial keyboard is provided in the MIKBUG routine at location $E1AC. When program control is passed to that routine and a key is pressed on a serial ASCII keyboard, the ASCII character will be returned in accumulator A. Another system input routine must be established in systems not using the 6800 microcomputer. Program control comes to this module from initializing the buffer or after converting an ASCII character to printer buffer control characters. Control always goes next to print when the ASCII character is a Carriage Return.

Print When . . .



Figure 8. "Print when ASCII is a carriage return"

The process to print when the ASCII character is a carriage return is illustrated in Figure 8. Step one tests the character and exits for all characters except the carriage return. This test is in read/write memory; all the rest is the LINPTR subroutine and can be in the read-only memory where it can be used as desired by a system designer to print the entire contents of a buffer full of control characters.

The printing process starts the motor and outputs the entire buffer contents under timing controlled by a data input line (or interrupt line) from the timing gear. The motor is then allowed to continue running until the reed switch closes to signal the end of line. If any of the timing signals fail to occur within a reasonable time, an exit will be made to system software to signal a failure. The system operator or system software must then determine what corrective action to take. This timeout facility is provided to prevent the printer program from entering an endless loop in read-only memory — an event which is difficult to diagnose.

This module uses the ASCII character in accumulator A, the buffer start address at BUFSTR, the buffer and address at BUFEND, the spot size control at SPTSIZ, and all the contents of the printer control buffer. The buffer will not have any words to cause printing until control has passed through the module described in Converting to ASCII Characters. A printed line will be output by this module when there are appropriate control characters in the buffer and the input character is a carriage return.

Control comes to this module when an ASCII character has been input. Control goes to convert the character to printer control words when not a carriage return; it goes to the system software when there is a late input from the timing sensor; and it goes to initialize the buffer when a line has been printed successfully.

This module contains the complete interface control software (more detailed description follows in the next four sections of Software Description).

Exit If ASCII Is Not Carriage Return



Figure 9. "Exit if ASCII is not a carriage return"

This part of the line typer software is illustrated in Figure 9. The printer control is started when the ASCII character is a carriage return or control goes to convert other characters to printer control words in the buffer.

Start the Motor

This part of the software is illustrated in Figure 10. The coding is part of the LINPTR subroutine which can be in read-only memory.



Figure 10. "Start the motor"

A binary word 0000 0000 is placed in the output register. This has the effect of moving bit seven to a low voltage in the interface to cause the motor to start. A time delay is necessary to allow the print head to be placed in contact with the paper. The delay is accomplished by waiting until the reed switch closes before proceeding to the next task. An exit is made to the system interrupt handler if the reed switch does not close.

Output the Buffer Contents

This part of the software controls the print electrode current. The illustration in Figure 11 shows the detail of operation.



Figure 11. "Output the buffer contents or timeout"

There are three loops in this module. The outer loop starts the column control character index at the start of the buffer as stored in BUFSTR. This loop ends at step 8 where a test is made to see if all column control characters have been output up to the address stored at BUFEND. When all have been output, the print electrodes are turned off and control goes to the next module.

The second loop controls spot size, starting at step 3 and ending at step 7. This loop counts timing marks as specified by SPTSIZ before allowing the column control character to be incremented.

The third loop is at steps 4, 5, and 6, where the timeout counter is decremented while waiting for a timing mark. If the timeout counter reaches zero before a timing mark is received, there will be an exit to the system software.

Stop At Line End



Figure 12. "Stop at line end"

This module is illustrated in Figure 12. It delays until the reed switch opens at line end and then puts a binary 1000 0000 in the output register to turn off the motor and print electrodes. A timeout loop is included to exit to system software in case the reed relay does not open within a reasonable time.

Convert ASCII Characters

The module to convert ASCII characters to printer control words is illustrated in Figure 13. This is a single subroutine called CACTPB in read/write memory. In code for the 6800 computer, the stack pointer is saved in the first step. It is used in transferring words from the conversion table to the print control buffer and is restored at step six to provide correct return to the control program.

Next, the ASCII character is tested to determine if it is in the printable set; nonprintable characters will be ignored. A printable ASCII character will be converted to five words for control of the print electrodes and moved to the buffer location indicated by the buffer storage pointer. Inter-character blank columns will be provided in the buffer by incrementing the buffer



Figure 13. "Convert ASCII character to printer control words"

storage pointer an additional number, as indicated by *BLANKN*, before storing the pointer for use the next time.

This routine uses the ASCII character in accumulator A, the table of conversion from ASCII to column control words, the buffer storage pointer at BUFPTR, and the spacing control at BLANKN. It updates the buffer control pointer, outputs control words to the buffer, and leaves a value called TBLCHR which points to the character moved from the conversion table.

Control comes to this routine from "print when ASCII is a carriage return" and goes to "input ASCII."

COMPUTER SOURCE CODE FOR THE 6800

Program listings are provided for 6800 computers in Listing 1. The comments consist of references to appropriate sections of the software description for detail of the functions being performed. A lookup table named TABLE provides CACTPB, the bit patterns for conversion of ASCII code to seven bit column control words. The Line Typer starts at address $3A5B; an additional entry point at $3A70 will allow entries via the "G" command to the MIKBUG monitor to stop the motor in case of a software interrupt and to restart at the proper address again by another "G" command.

The memory map for this code after assembly is provided in Table 2.

Table 2. Memory Map for 6800

3FB4	top of 44 character buffer
3E80	beginning of buffer
3DFF	top of table
3CC0	start of table
3C7C	LINPTR
3C4F	INITPR
3C2E	BEQOTO
3C0D	BNEOTO
3C06	XDELAY
3C00	DELAY
3B81	top of auxiliary routines
3B70	GSTART
3B5B	TESTPR
3B1F	CACTPB
3B00	CAINIT

SUMMARY

The printer is extremely flexible and works well in just about any system configuration. Although only software for the 6800 was shown, the techniques discussed and demonstrated work with 8080 type machines.□

Program Follows

Program Listing

```
00001                    NAM     TABLE
00002                    OPT     MEM
00003 3CC0               ORG     $3CC0
00004 3CC0 0000          FDB     0
00005 3CC2 0000          FDB     0
00006 3CC4 00            FCB     0          * SPACE
00007 3CC5 0000          FDB     0
00008 3CC7 4F00          FDB     $4F00
00009 3CC9 00            FCB     0          * !
00010 3CCA 0007          FDB     7
00011 3CCC 0007          FDB     7
00012 3CCE 00            FCB     0          * "
00013 3CCF 147F          FDB     $147F
00014 3CD1 147F          FDB     $147F
00015 3CD3 14            FCB     $14        * #
00016 3CD4 242A          FDB     $242A
00017 3CD6 7F2A          FDB     $7F2A
00018 3CD8 12            FCB     $12        * $
00019 3CD9 2313          FDB     $2313
00020 3CDB 0864          FDB     $864
00021 3CDD 62            FCB     $62        * %
00022 3CDE 304E          FDB     $304E
00023 3CE0 5926          FDB     $5926
00024 3CE2 50            FCB     $50        * &
00025 3CE3 0004          FDB     4
00026 3CE5 0201          FDB     $201
00027 3CE7 00            FCB     0          * '
00028 3CE8 001C          FDB     $1C
00029 3CEA 2241          FDB     $2241
00030 3CEC 00            FCB     0          * (
00031 3CED 0041          FDB     $41
00032 3CEF 221C          FDB     $221C
00033 3CF1 00            FCB     0          * )
00034 3CF2 1408          FDB     $1408
00035 3CF4 3E08          FDB     $3E08
00036 3CF6 14            FCB     $14        * *
00037 3CF7 0808          FDB     $808
00038 3CF9 3E08          FDB     $3E08
00039 3CFB 08            FCB     8          * +
00040 3CFC 0040          FDB     $40
00041 3CFE 3000          FDB     $3000
00042 3D00 00            FCB     0          * ,
00043 3D01 0808          FDB     $808
00044 3D03 0808          FDB     $808
00045 3D05 08            FCB     8          * -
00046 3D06 0060          FDB     $60
00097 3D5B 0201          FDB     $201
00098 3D5D 7109          FDB     $7109
00099 3D5F 06            FCB     6          * ?
00100 3D60 3E41          FDB     $3E41
00101 3D62 5D55          FDB     $5D55
00102 3D64 5E            FCB     $5E        * @
00103 3D65 7E09          FDB     $7E09
00104 3D67 0909          FDB     $909
00105 3D69 7E            FCB     $7E        * A
00106 3D6A 7F49          FDB     $7F49
00107 3D6C 4949          FDB     $4949
00108 3D6E 36            FCB     $36        * B
00109 3D6F 3E41          FDB     $3E41
00110 3D71 4141          FDB     $4141
00111 3D73 22            FCB     $22        * C
00112 3D74 7F41          FDB     $7F41
00113 3D76 4122          FDB     $4122
00114 3D78 1C            FCB     $1C        * D
00115 3D79 7F49          FDB     $7F49
00116 3D7B 4949          FDB     $4949
00117 3D7D 41            FCB     $41        * E
00118 3D7E 7F09          FDB     $7F09
00119 3D80 0909          FDB     $909
00120 3D82 01            FCB     1          * F
00121 3D83 3E41          FDB     $3E41
00122 3D85 4149          FDB     $4149
00123 3D87 79            FCB     $79        * G
00124 3D88 7F08          FDB     $7F08
00125 3D8A 0808          FDB     $808
00126 3D8C 7F            FCB     $7F        * H
00127 3D8D 0041          FDB     $41
00128 3D8F 7F41          FDB     $7F41
00129 3D91 00            FCB     0          * I
00130 3D92 2040          FDB     $2040
00131 3D94 413F          FDB     $413F
00132 3D96 01            FCB     1          * J
00133 3D97 7F08          FDB     $7F08
00134 3D99 1422          FDB     $1422
00135 3D9B 41            FCB     $41        * K
00136 3D9C 7F40          FDB     $7F40
00137 3D9E 4040          FDB     $4040
00138 3DA0 40            FCB     $40        * L
00139 3DA1 7F02          FDB     $7F02
00140 3DA3 0402          FDB     $402
00141 3DA5 7F            FCB     $7F        * M
00142 3DA6 7F04          FDB     $7F04
00143 3DA8 0810          FDB     $810
00144 3DAA 7F            FCB     $7F        * N
00145 3DAB 3E41          FDB     $3E41
00146 3DAD 4141          FDB     $4141
```

```
00047 3D08 6000          FDB     $6000
00048 3D0A 00            FCB     0          * .
00049 3D0B 2010          FDB     $2010
00050 3D0D 0804          FDB     $804
00051 3D0F 02            FCB     2          * /
00052 3D10 3E51          FDB     $3E51
00053 3D12 4945          FDB     $4945
00054 3D14 3E            FCB     $3E        * 0
00055 3D15 0042          FDB     $42
00056 3D17 7F40          FDB     $7F40
00057 3D19 00            FCB     0          * 1
00058 3D1A 4261          FDB     $4261
00059 3D1C 5149          FDB     $5149
00060 3D1E 46            FCB     $46        * 2
00061 3D1F 2141          FDB     $2141
00062 3D21 454B          FDB     $454B
00063 3D23 31            FCB     $31        * 3
00064 3D24 1814          FDB     $1814
00065 3D26 127F          FDB     $127F
00066 3D28 10            FCB     $10        * 4
00067 3D29 2745          FDB     $2745
00068 3D2B 4545          FDB     $4545
00069 3D2D 39            FCB     $39        * 5
00070 3D2E 3C4A          FDB     $3C4A
00071 3D30 4949          FDB     $4949
00072 3D32 30            FCB     $30        * 6
00073 3D33 0171          FDB     $171
00074 3D35 0905          FDB     $905
00075 3D37 03            FCB     $3         * 7
00076 3D38 3649          FDB     $3649
00077 3D3A 4949          FDB     $4949
00078 3D3C 36            FCB     $36        * 8
00079 3D3D 0649          FDB     $649
00080 3D3F 4929          FDB     $4929
00081 3D41 1E            FCB     $1E        * 9
00082 3D42 0063          FDB     $63
00083 3D44 6300          FDB     $6300
00084 3D46 00            FCB     0          * :
00085 3D47 0000          FDB     0
00086 3D49 4033          FDB     $4033
00087 3D4B 00            FCB     0          * ;
00088 3D4C 0814          FDB     $814
00089 3D4E 2241          FDB     $2241
00090 3D50 00            FCB     0          * <
00091 3D51 1414          FDB     $1414
00092 3D53 1414          FDB     $1414
00093 3D55 14            FCB     $14        * =
00094 3D56 0041          FDB     $41
00095 3D58 2214          FDB     $2214
00096 3D5A 08            FCB     8          * >
```

```
00147 3DAF 3E            FCB     $3E        * O
00148 3DB0 7F09          FDB     $7F09
00149 3DB2 0909          FDB     $909
00150 3DB4 06            FCB     6          * P
00151 3DB5 3E41          FDB     $3E41
00152 3DB7 5121          FDB     $5121
00153 3DB9 5E            FCB     $5E        * Q
00154 3DBA 7F09          FDB     $7F09
00155 3DBC 1929          FDB     $1929
00156 3DBE 46            FCB     $46        * R
00157 3DBF 4649          FDB     $4649
00158 3DC1 4949          FDB     $4949
00159 3DC3 31            FCB     $31        * S
00160 3DC4 0101          FDB     $101
00161 3DC6 7F01          FDB     $7F01
00162 3DC8 01            FCB     1          * T
00163 3DC9 7F40          FDB     $7F40
00164 3DCB 4040          FDB     $4040
00165 3DCD 7F            FCB     $7F        * U
00166 3DCE 1F20          FDB     $1F20
00167 3DD0 4020          FDB     $4020
00168 3DD2 1F            FCB     $1F        * V
00169 3DD3 3F40          FDB     $3F40
00170 3DD5 3840          FDB     $3840
00171 3DD7 3F            FCB     $3F        * W
00172 3DD8 6314          FDB     $6314
00173 3DDA 0814          FDB     $814
00174 3DDC 63            FCB     $63        * X
00175 3DDD 0708          FDB     $708
00176 3DDF 7008          FDB     $7008
00177 3DE1 07            FCB     7          * Y
00178 3DE2 6151          FDB     $6151
00179 3DE4 4945          FDB     $4945
00180 3DE6 43            FCB     $43        * Z
00181 3DE7 007F          FDB     $7F
00182 3DE9 4141          FDB     $4141
00183 3DEB 00            FCB     0          * [
00184 3DEC 0204          FDB     $204
00185 3DEE 0810          FDB     $810
00186 3DF0 20            FCB     $20        * \
00187 3DF1 0041          FDB     $41
00188 3DF3 417F          FDB     $417F
00189 3DF5 00            FCB     0          * ]
00190 3DF6 0402          FDB     $402
00191 3DF8 7F02          FDB     $7F02
00192 3DFA 04            FCB     4          * ^
00193 3DFB 081C          FDB     $81C
00194 3DFD 2A08          FDB     $2A08
00195 3DFF 00            FCB     0          * _
00196                    END
```

```
00540 3A35 C9 00          ADC B  #$00
00550 3A37 97 29          STA A  TBLCHR+1
00560 3A39 D7 28          STA B  TBLCHR
00570 3A3B 9E 28          LDS    TBLCHR
00580 3A3D DE 20          LDX    BUFPTR
00590 3A3F C6 05          LDA B  #$05
00600 3A41 32             PUL A
00610 3A42 A7 00          STA A  0,X
00620 3A44 08             INX
00630 3A45 5A             DEC B
00640 3A46 26 F9          BNE    *-5
00645                *1.5.3***
00660 3A48 D6 2E          LDA B  BLANKN
00680 3A4A 08             INX
00690 3A4B 5A             DEC B
00700 3A4C 26 FC          BNE    *-2
00705                *1.5.4***
00710 3A4E DF 20          STX    BUFPTR
00720 3A50 9C 2C          CPX    BUFEND
00730 3A52 26 04          BNE    CAEXIT
00740 3A54 DE 2A          LDX    BUFSTR
00750 3A56 DF 20          STX    BUFPTR
00760 3A58 9E 24    CAEXIT LDS   SCRTCH
00770 3A5A 39             RTS
00775                *TO 1.1***
00800 3A5B BD 3C4F  TESTPR JSR   INITPR
00805                *TO 1.2***
00810 3A5E 8D A0          BSR    CAINIT
00815                *1.3*****
00820 3A60 BD E1AC  INCHAR JSR   $E1AC
00830 3A63 81 0D          CMP A  #$0D
00835                *1.4.1**
00840 3A65 26 05          BNE    *+7
00845                *TO 1.4.2*
00850 3A67 BD 3C7C        JSR    LINPTR
00860 3A6A 20 F2          BRA    TESTPR+3
00865                *TO 1.5**
00870 3A6C 8D B1          BSR    CACTPR
00880 3A6E 20 F0          BRA    INCHAR
00890 3A70 CE 3A7A  GSTART LDX   #SWIEND
00900 3A73 DF 6C          STX    $6C
00910 3A75 BD 3A5B        JSR    TESTPR
00920 3A78 20 F6          BRA    GSTART
00930                * SET A048 TO GSTART FOR MONITOR ENTRY
00940 3A7A 86 80    SWIEND LDA A  #$80
00950 3A7C B7 8006        STA A  PTRDR
00960 3A7F 7E E0E0        JMP    $E0E0
01000 3C00                ORG    $3C00
01010      006A     INTVCT EQU    $006A
01020 3C00 C6 FF    DELAY  LDA B  #$FF
01030 3C02 5A             DEC B
01040 3C03 26 FD          BNE    *-1
01050 3C05 39             RTS
01060 3C06 8D 05    XDELAY BSR    BNEOTO
01070 3C08 8D F6          BSR    DELAY
01080 3C0A 8D 22          BSR    BEQOTO
01090 3C0C 39             RTS
01100 3C0D 86 14    BNEOTO LDA A  #$14
01110 3C0F B7 8007        STA A  PTRCR
01120 3C12 B6 8006        LDA A  PTRDR
01130 3C15 C6 FF          LDA B  #$FF
01140 3C17 D7 24          STA B  SCRTCH
01150 3C19 D7 25          STA B  SCRTCH+1
01160 3C1B B6 8007        LDA A  PTRCR
01170 3C1E 84 40          AND A  #$40
01180 3C20 26 0B          BNE    *+13
01190 3C22 7A 0025        DEC    SCRTCH+1
01200 3C25 26 F4          BNE    *-10
01210 3C27 7A 0024        DEC    SCRTCH
01220 3C2A 26 ED          BNE    *-17
01230 3C2C 3F             SWI
01240 3C2D 39             RTS
01250 3C2E 86 04    BEQOTO LDA A  #$04
01260 3C30 B7 8007        STA A  PTRCR
01270 3C33 B6 8006        LDA A  PTRDR
01280 3C36 C6 FF          LDA B  #$FF
01290 3C38 D7 24          STA B  SCRTCH
01300 3C3A D7 25          STA B  SCRTCH+1
01310 3C3C B6 8007        LDA A  PTRCR
01320 3C3F 84 40          AND A  #$40
01330 3C41 27 0B          BEQ    *+13
01340 3C43 7A 0025        DEC    SCRTCH+1
01350 3C46 26 F4          BNE    *-10
01360 3C48 7A 0024        DEC    SCRTCH
01370 3C4B 26 ED          BNE    *-17
01380 3C4D 3F             SWI
01390 3C4E 39             RTS
01395                *1.1*****
01400 3C4F 8D 00    INITPR BSR    *+2
01410 3C51 32             PUL A
01420 3C52 33             PUL B
01425                *1.1.1**
01420 3C53 CB 6E          ADD B  #$6E
01430 3C55 97 26          STA A  TBLADR
01440 3C57 D7 27          STA B  TBLADR+1
01445                *1.1.2***
01450 3C59 7F 8007        CLR    PTRCR
01460 3C5C 86 FF          LDA A  #$FF
01470 3C5E B7 8006        STA A  PTRDR
01480 3C61 86 04          LDA A  #$04
01490 3C63 B7 8007        STA A  PTRCR
01495                *1.1.3**
01500 3C66 86 00          LDA A  #$00
01510 3C68 B7 8006        STA A  PTRDR
01520 3C6B 8D 99          BSR    XDELAY
01525                *1.1.4**
01530 3C6D 8D 49          BSR    LINEND+3
01535                *1.1.5***
```

```
00010                          NAM    PRDRVR
00015                  * COPYRIGHT 1978 EPA  1/16/78
00020 3A00                     ORG    $3A00
00030                          OPT    MEM
00040       0022       CHRCNT EQU    $22
00050       0024       SCRTCH EQU    $24
00060       0023       SPTSIZ EQU    $23
00070       0026       TBLADR EQU    $26
00080       0028       TBLCHR EQU    $28
00090       002A       BUFSTR EQU    $2A
00100       0020       BUFPTR EQU    $20
00110       002C       BUFEND EQU    $2C
00120       002E       BLANKN EQU    $2E
00130       3E80       BUFLOC EQU    $3E80
00140       8006       PTRDR  EQU    $8006
00150       8007       PTRCR  EQU    $8007
00195                  *1.2*****
00200 3A00 CE 3E80     CAINIT LDX    #BUFLOC
00205                  *1.2.1***
00210 3A03 DF 2A               STX    BUFSTR
00220 3A05 DF 20               STX    BUFPTR
00230 3A07 96 22               LDA A  CHRCNT
00240 3A09 97 24               STA A  SCRTCH
00250 3A0B 86 00               LDA A  #$00
00260 3A0D D6 2E               LDA B  BLANKN
00270 3A0F CB 05               ADD B  #$05
00275                  *1.2.2***
00280 3A11 A7 00               STA A  0,X
00290 3A13 08                  INX
00300 3A14 5A                  DEC B
00310 3A15 26 FA               BNE    *-4
00320 3A17 7A 0024             DEC    SCRTCH
00330 3A1A 26 F1               BNE    *-13
00335                  *1.2.3***
00340 3A1C DF 2C               STX    BUFEND
00350 3A1E 39                  RTS
00355                  *1.5*****
00400 3A1F 9F 24       CACTPB  STS    SCRTCH
00410 3A21 80 20               SUB A  #$20
00415                  *1.5.1***
00420 3A23 2D 33               BLT    CAEXIT
00430 3A25 81 3F               CMP A  #$3F
00435                  *1.5.1***
00440 3A27 2E 2F               BGT    CAEXIT
00445                  *1.5.2***
00450 3A29 0C                  CLC
00460 3A2A 16                  TAB
00470 3A2B 48                  ASL A
00480 3A2C 48                  ASL A
00490 3A2D 1B                  ABA
00500 3A2E D6 26               LDA B  TBLADR
00510 3A30 C9 00               ADC B  #$00
00520 3A32 0C                  CLC
00530 3A33 9B 27               ADD A  TBLADR+1
01540 3C6F 86 01               LDA A  #$01
01550 3C71 97 23               STA A  SPTSIZ
01560 3C73 86 2C               LDA A  #$2C
01570 3C75 97 22               STA A  CHRCNT
01580 3C77 86 02               LDA A  #$02
01590 3C79 97 2E               STA A  BLANKN
01600 3C7B 39                  RTS
01700 3C7C 0F          LINPTR  SEI
01705                  *1.4.2**
01710 3C7D 86 00               LDA A  #$00
01720 3C7F B7 8006             STA A  PTRDR
01730 3C82 8D 82               BSR    XDELAY
01735                  *1.4.3.1*
01740 3C84 DE 2A               LDX    BUFSTR
01745                  *1.4.3.2*
01750 3C86 A6 00       LINLUP  LDA A  0,X
01760 3C88 B7 8006             STA A  PTRDR
01765                  *1.4.3.3*
01770 3C8B D6 23               LDA B  SPTSIZ
01780 3C8D 37          CHRLUP  PSH B
01790 3C8E B6 8006             LDA A  PTRDR
01795                  *1.4.3.4*
01800 3C91 C6 0F               LDA B  #$0F
01810 3C93 D7 24               STA B  SCRTCH
01820 3C95 D7 25               STA B  SCRTCH+1
01830 3C97 B6 8007             LDA A  PTRCR
01835                  *1.4.3.5*
01840 3C9A 2B 0B               BMI    *+13
01845                  *1.4.3.6*
01850 3C9C 7A 0025             DEC    SCRTCH+1
01860 3C9F 26 F6               BNE    *-8
01870 3CA1 7A 0024             DEC    SCRTCH
01880 3CA4 26 EF               BNE    *-15
01890 3CA6 3F                  SWI
01895                  *1.4.3.7*
01900 3CA7 33                  PUL B
01910 3CA8 5A                  DEC B
01920 3CA9 26 E2               BNE    CHRLUP
01925                  *1.4.3.8*
01930 3CAB 08                  INX
01940 3CAC 9C 2C               CPX    BUFEND
01950 3CAE 26 D6               BNE    LINLUP
01955                  *1.4.3.9*
01960 3CB0 86 00               LDA A  #$00
01970 3CB2 B7 8006             STA A  PTRDR
01975                  *1.4.4.1*
01980 3CB5 BD 3C0D     LINEND  JSR    BNEOTO
01985                  *1.4.4.5*
01990 3CB8 86 80               LDA A  #$80
02000 3CBA B7 8006             STA A  PTRDR
02010 3CBD 0E                  CLI
02020 3CBE 39                  RTS
02030                          END
```